VBAZugriff auf das OutlookAdressbuch
 Teil 1
Versionen: Access 2007, 2003, 2002/XP und 2000

Für den Zugriff auf das Adressbuch stellt das Outlook-Objektmodell keine Möglichkeiten bereit. Um eine Adressauswahl zu realisieren ist deshalb ein Umweg über CDO (Collaboration Data Objects) notwendig. Dort steht eine Funktion "AddressBook" zur Verfügung, die als Ergebnis eine "Recipients"-Auflistung mit den gewählten Empfängern liefert. Wie genau diese Möglichkeiten per VBA genutzt werden können, erläutert dieser Tipp der Woche.

Voraussetzung für die Nutzung dieser Funktion ist zunächst, dass CDO auf dem System installiert ist. Prüfen können Sie dies, indem Sie die VBA-Entwicklungsumgebung öffnen, dort Extras-Verweise anwählen und nach einem Eintrag "Microsoft CDO 1.2x Object Library" suchen. Ist der Eintrag nicht vorhanden, gehen Sie bei Access 2003, 2002/XP oder 2000 wie folgt vor, um CDO nachträglich zu installieren:

1. Beenden Sie Outlook und alle anderen, eventuell laufenden Office-Anwendungen.

2. Legen Sie die Office-CD ein und starten Sie dort "Setup".

3. Klicken Sie auf die Schaltfläche Features Hinzufügen/entfernen.

4. Wenn Sie Office 2003 einsetzen, aktivieren Sie die Option Erweiterte Anpassung Von Anwendungen und klicken Sie dann auf Weiter.

5. Aktivieren Sie im Bereich "Microsoft Outlook für Windows" für den Eintrag "Collaboration Data Objects" die Option Vom Arbeitsplatz Starten.

6. Klicken Sie auf die Schaltfläche Jetzt Aktualisieren und verlassen Sie das Setup anschließend wieder.

Bei Access 2007 wird CDO nicht mehr auf der CD mitgeliefert, Sie können das Setup aber kostenlos unter folgender Adresse herunterladen:

	
	 Kostenloser Download: Collaboration Data Objects



Beenden Sie Outlook, starten Sie das CDO-Setup nach dem Download per Doppelklick und folgen Sie den weiteren Anweisungen des Installationsprogramms.

Danach können Sie Outlook wieder starten, in der VBA-Entwicklungsumgebung ist nun unter Extras-Verweise der Eintrag "Microsoft CDO 1.2x Object Library" vorhanden und kann über das Kontrollkästchen aktiviert werden.

Um das Adressbuch für die Auswahl einer oder mehrere Adressen anzeigen zu lassen, kopieren Sie zunächst folgende Deklarationen und die Funktion "AddrBookTest()" in ein vorhandenes oder in ein neues Modul Ihrer Datenbank:

Option Explicit

Const txtTitel = "AddrBookTest"

Function AddrBookTest(OneAddress As Boolean)
  Dim objCDO As MAPI.Session
  Dim objRecs As MAPI.Recipients
  Dim strResult As String, strErr As String, I&

  On Error Resume Next
  AddrBookTest = "" 'Default: Nichts ausgewählt
  Set objCDO = CreateObject("MAPI.Session")
  objCDO.Logon "", "", False, False
  If Err <> 0 Or objCDO Is Nothing Then
    Beep
    MsgBox "Fehler bei Initialisierung " & _
           "der CDO-Schnittstelle: " & _
           Err.Description, _
           vbOKOnly + vbExclamation, txtTitel
    GoTo Exit_Func
  End If

  With objCDO.Session
    Err = 0
    Set objRecs = .AddressBook(Title:="Empfänger wählen:")
    DoEvents
    If Err <> 0 Then
      strErr = Err.Description
      If InStr(LCase$(strErr), "user_cancel") = 0 Then
       Beep
       MsgBox "Problem beim Aufruf des Adressbuches (" & _
               strErr & ")...", _
               vbOKOnly + vbExclamation, txtTitel
       GoTo Exit_Func
     End If 'User_Cancel?
    End If 'Err<> 0...
  End With
  If objRecs Is Nothing Then Exit Function

  Err = 0
  strResult = ""
  For I = 1 To objRecs.Count
    With objRecs.Item(I)
     strResult = strResult & Trim$(.AddressEntry.Name) & _
     "|" & Trim$(.AddressEntry.Address) & vbCrLf
    End With
    If OneAddress And I = 1 Then Exit For
  Next I

  If Err <> 0 Then
    strErr = Err.Description
    Beep
    MsgBox "Problem bei der Übernahme der Adresse (" & _
           strErr & ")...", _
           vbOKOnly + vbExclamation, txtTitel
  Else 'letztes vbCrLf wieder raus
    strResult = Left$(strResult, Len(strResult) - 2)
    AddrBookTest = strResult
  End If

Exit_Func:
  objCDO.Logoff

  Set objRecs = Nothing
  Set objCDO = Nothing

End Function

Wie diese Funktion im Einzelnen arbeitet und das Funktionsergebnis ausgewertet wird, erläutern wir im in der nächsten Ausgabe.
VBA-Zugriff auf das Outlook-Adressbuch, Teil 2
Versionen: Access 2007, 2003, 2002/XP und 2000

Im ersten Teil dieses Tipps der Woche hatten wir zunächst die Installation von CDO erläutert und die Basisfunktion für den Aufruf der Adressbuch-Auswahl vorgestellt. In dieser Ausgabe geht es nun um die Funktionsweise und die Auswertung der Rückmeldung.

Die Funktion " AddrBookTest()" initialisiert zunächst eine CDO-Session mit dem Standard-Profil, fängt dabei einen eventuell auftretenden Fehler ab und zeigt gegebenenfalls eine entsprechende Meldung an. Über diese Session kann die Funktion "AddressBook" aufgerufen werden, die als Ergebnis eine "Recipients"-Auflistung liefert. Eventuell auftretende Fehler untersuchen wir auf die Zeichenfolge "user_cancel", die einen Klick auf Abbrechen signalisiert. Alle anderen Fehler werden entsprechend per Meldung zur Kenntnis gegeben, bevor die Routine verlassen wird. Die Eigenschaft "Count" der "Recipients"-Auflistung gibt Auskunft darüber, wie viele Einträge ausgewählt wurden. Dementsprechend gehen wir in einer Schleife alle Einträge durch und setzen diese zu einem Funktionsergebnis zusammen, das zeilenweise Namen und Adressen durch das Pipe-Zeichen getrennt enthält. Wenn der Parameter "OneAddress" mit "True" angegeben war, verlassen wir die Schleife nach Bearbeitung des ersten Eintrages. Einen eventuell auftretenden Fehler, beispielsweise ausgelöst durch einen Klick auf Nein in der bei Zugriffen auf E-Mail-Adressen angezeigten Sicherheitsabfrage von Outlook, fangen wir entsprechend ab und zeigen ihn per Meldung an.

Das Funktionsergebnis ist entweder eine Zeichenfolge wie zum Beispiel "Vorname Nachname|nachname@provider.de" (OneAddress= True), also eine Kombination aus dem Namen und der E-Mail-Adresse oder eine per "vbCrLf" getrennte Liste oben genannten Aufbaus (OneAddress= False). Um auf die einzelnen Einträge der Liste zuzugreifen, wandeln Sie sie zunächst per "Split" in ein Array um und zerlegen Sie dann anhand des Trenners "|" in Name und E-Mail-Adresse:

Sub Test_1()
  Dim X As Variant, I&
  Dim strListe As String
  Dim strName As String
  Dim strEMail As String

  strListe = AddrBookTest(False)
  X = Split(strListe, vbCrLf)
  For I = 0 To UBound(X)
    strName = Left$(X(I), InStr(X(I), "|") - 1)
    strEMail = Mid$(X(I), InStr(X(I), "|") + 1)
    Debug.Print strName, strEMail
  Next I

End Sub

Eine einzelne Adresse zerlegen Sie wie oben gezeigt per "Left$()" und "Mid$()" in Ihre Bestandteile:

Sub Test_2()
  Dim strAusw As String
  Dim strName As String
  Dim strEMail As String

  strAusw = AddrBookTest(True) 'Eine Adresse
  strName = Left$(strAusw, InStr(strAusw, "|") - 1)
  strEMail = Mid$(strAusw, InStr(strAusw, "|") + 1)
  Debug.Print strName, strEMail

End Sub

Um eine Adresse in einem Formular auswählen zu lassen, legen Sie eine Schaltfläche "btnChooseAdr" oder ähnlich mit folgender Ereignisprozedur "Beim Klicken" an:

Sub btnChooseAdr_Click()
  Dim strAusw As String
  Dim strName As String
  Dim strEMail As String

  strAusw = AddrBookTest(True)
  If strAusw <> "" Then
    strName = Left$(strAusw, InStr(strAusw, "|") - 1)
    strEMail = Mid$(strAusw, InStr(strAusw, "|") + 1)
    Me.EMailAdr = strName & " (" & strEMail & ")"
  End If

End Sub

Die im Adressbuch erfolgte Auswahl wird zunächst in "strAusw" festgehalten und anschließend per "Left$()" und "Mid$()" in die Bestandteile "Name" und "Adresse" getrennt. Diese Daten werden dann per Zuweisung ins Feld " EMailAdr" im Format "Vorname Nachname (Adresse)" geschrieben.
